

# Using 22 Easy Tricks with ODS to Generate Colorful Reports

Ben Cochran, The Bedford Group, Raleigh, NC

## ABSTRACT

There are many ways to create reports with the SAS® System. This can be done with any number of Procedures, the DATA step, as well as products such as Enterprise Guide, and the Web Report Studio. The appearance of these reports can be enhanced by using the many features of the Output Delivery System (ODS). ODS is so robust and has so many ways to enhance reports, that one can spend a great deal of time mastering these skills. This paper will show the user some fairly simple techniques to gain the benefits of ODS. The approach taken here is to show the user some familiar looking report generating code, and then add some very simple options that will have a big impact on the output.

## INTRODUCTION

ODS offers the users a way to send the output of their SAS programs to a destination other than the default one, in most cases this means the output window. These different destinations also offer some aesthetic features that really enhance the appearance of the reports. This paper is intended for users who have not worked with ODS, or at least not much. lately. First, a look is taken at the general syntax, and then a series of reports will be generated starting with a very simple report and then working up to more complex reports.

## TYPICAL SYNTAX

To begin using ODS, all you have to do is 'sandwich' your PROC step with two simple ODS statements, one to specify the output location, and the other to close the location.

**ODS destination-type destination;**

```
PROC procedure data= SAS data set options ;
    ... ;
    ... ;
RUN;
```

**ODS destination-type CLOSE;**

Output destinations can be specified as HTML, RTF, PDF, Excel spreadsheets as well as others.

**Task 1:** Generate a basic Report using the PRINT procedure, and send it to an HTML file. Use the first statement to specify an HTML location, and use the second one to close it.

**ods html file = 'c:\reports.html';**

**title 'Options Example';**

**options nodate nocenter**

**PROC PRINT data = sashelp.class(rename=(sex=gender) drop=height  
obs=7 firstobs=3 ) ;**

**RUN;**

**ods html CLOSE;**

## Options Example

Obs	Name	Gender	Age	Weight
3	Barbara	F	13	98.0
4	Carol	F	14	102.5
5	Henry	M	14	102.5
6	James	M	12	83.0
7	Jane	F	12	84.5

Output 1 – Default ODS output with PROC PRINT.

The output shows the default attributes for this destination, such as foreground and background color, font color and size, etc. There is a fairly easy way to control these attributes, and that is what this paper is all about. There is an option known as the STYLE= option that can control many aspects of your output. The general syntax is:

**STYLE = { attribute – 1 = value – 1 ...  
attribute – n = value – n } ;**

**or**

**STYLE (location) = {attribute = value}**

The STYLE= option can be used with many report generating procedures in SAS and can be abbreviated as S=. First, let's look at using this option with PROC PRINT.

Location (Statement)	Report Location Modified	Can also be used on these statements
BYLABEL	Label for the BY variable on line containing the SUM totals	None
DATA	The data cells for all columns	ID, SUM, VAR
GRANDTOTAL	The SUM line containing the grand total for the whole report	SUM
HEADER	All column headings	ID, SUM, VAR
N	Text on the N= line	None
OBS	The data in the OBS column	None
OBSHEADER	The header of the OBS column	None
TABLE	The structural part of the report (border, widths, space between cells)	None
TOTAL	The SUM line containing totals for each BY group	SUM

Looking at this in more detail, let us graphically see what this looks like in output created from the SASHELP.CLASS dataset. Some examples of placing the STYLE= option on the PROC PRINT step include...

On the **PROC PRINT** statement:

- style (obsheader=cyan) →
- style (data)= {background=yellow} →
- style (obs) = {background=pink} →
- style (total) = {background=blue} →

Sex=F				
Obs	Name	Age	Height	Weight
6	Carol	14	62.8	102.5
7	Judy	14	64.3	90.0
8	Janet	15	62.5	112.5
9	Mary	15	66.5	112.0
Sex		58		

### Trick 2:

Now, let's put this to work in a PROC PRINT step. Use the STYLE = option to control the background color of the column headers.

```
options ls=64 center pageno=1 nodate;
title 'Cargo Revenue Generated on Lisbon Flights';

ods pdf file='c:\Lisbon.pdf';
proc print data=sas_1.mdv(rename=( _4cast2006=sales2006))
  style (header)= {background=yellow};
  where city = 'LISBON';
  var city origcity code type sales2005 sales2006;
run;
ods pdf close;
```

**Program 2.** Controlling the background color of the column headers.

<i>Cargo Revenue Generated on Lisbon Flights</i>							1
Obs	CITY	ORIGCITY	CODE	TYPE	Sales2005	sales2006	
71	LISBON	FRANKFRT	NEXT DAY	DC10	4,223.00	4,514.58	
72	LISBON	SYDNEY	THIRD DAY	747	3,281.00	3,889.67	
73	LISBON	NEW YORK	SECOND DAY	777	154.50	158.76	
74	LISBON	SAN FRAN	NEXT DAY	777	64.50	72.28	
75	LISBON	NEW YORK	NEXT DAY	777	76.00	82.25	
76	LISBON	SYDNEY	THIRD DAY	777	54.00	55.97	

**Output 2.**

**Trick 3:**

Write a PROC PRINT step to Control the size of the font on the TOTAL line and change the background color of the column headings..

```
options ls=74 ps = 22 center pageno=1 nodate;
title 'Cargo Revenue Generated on Flights Lisbon Flights';

ods pdf file='c:\Lisbon.pdf';
proc print data=sas_1.mdv(rename=(_4cast2006=sales2006))
  style (header)= {background=cyan} noobs
  style (grandtotal) = {background=grayee
                        font=(Arial) font_size=6} ;
  where city = 'LISBON';
  var origcity code type sales2005 sales2006;
  sum sales2005 sales2006;
run;
ods pdf close;
```

**Program 3.**

Notice the introduction of the GRANDTOTAL location and the FONT attribute.

<i><b>Cargo Revenue Generated on Lisbon Flights</b></i>				
ORIGCITY	CODE	TYPE	Sales2005	sales2006
FRANKFRT	NEXT DAY	DC10	4,223.00	4,514.58
SYDNEY	THIRD DAY	747	3,281.00	3,889.67
NEW YORK	SECOND DAY	777	154.50	158.76
SAN FRAN	NEXT DAY	777	64.50	72.28
NEW YORK	NEXT DAY	777	76.00	82.25
SYDNEY	THIRD DAY	777	54.00	55.97
			7,853.00	8,773.51

**Output 3.****Trick 4.**

Enhance the report by assigning a labels, adding a second title, and color the background of the report a light shade of gray. The OPTIONS and TITLE statements are shown below. Notice the use of the NODATE option to suppress the date from appearing in the default location in the upper right hand corner of the page. Also notice the use of the &SYSDATE9 macro variable in the second title. This allows the user to control where the date appears in the output.

```
options ls=74 ps = 22 center pageno=1 nodate;
title 'Cargo Revenue Generated on Lisbon Flights';
title2 "Report Generated on &sysdate9";
```

Program 4a.

```
ods pdf file='c:\Lisbon.pdf';
proc print data=sas_1.mdv(rename=( _4cast2006=sales2006))
  style (data) = {background=grayee}
  style (header)= {background=yellow} noobs split='*'
  style (grandtotal) = {background=white font=(Arial) font_size=6};
  where city = 'LISBON';
  var origcity code type sales2005 sales2006;
  sum sales2005 sales2006;
  format sales2005 sales2006 dollar10.2;
  label sales2005 = '2005 * Sales'
        sales2006 = '2006 * Sales'
        city      = 'Destination * City'
        origcity  = 'Originating * City'
        code      = 'Delivery * Type'
        type      = 'Aircraft * Type';
run;
ods pdf close;
```

Program 4b.

Cargo Revenue Generated on Flights Lisbon Flights Report Generated on 03JAN2007						1
Destination City	Originating City	Delivery Type	Aircraft Type	2005 Sales	2006 Sales	
LISBON	FRANKFRT	NEXT DAY	DC10	\$4,223.00	\$4,514.58	
LISBON	SYDNEY	THIRD DAY	747	\$3,281.00	\$3,889.67	
LISBON	NEW YORK	SECOND DAY	777	\$154.50	\$158.76	
LISBON	SAN FRAN	NEXT DAY	777	\$64.50	\$72.28	
LISBON	NEW YORK	NEXT DAY	777	\$76.00	\$82.25	
LISBON	SYDNEY	THIRD DAY	777	\$54.00	\$55.97	
				\$7,853.00	\$8,773.51	

Output 4.

#### Trick 5:

Create a SUBTOTAL row and give it a different color from the rest of the report.

```
ods pdf file='c:\Lisbon.pdf';
proc print data=texas split='*' noobs
  style (data) = {background=grayee }
  style (header)= {background=yellow foreground=blue font=(Arial)
                  font_size=6}
  style (total) = {background=light green}
  style (grandtotal) = {background=white
                      font=(Arial) font_size=6} ;
```

```

by origcity;
id origcity city;
where origcity in('NEW YORK', 'SAN FRAN') and
       city in('AUSTIN', 'DALLAS');
var sales2005 sales2006;
sum sales2005 sales2006;
run;

ods pdf close;

```

### Program 5.

Notice all the STYLE options on the PROC PRINT statement.

#### *Cargo Revenue Generated on Texas Flights Originating from the Coast*

Originating City	Destination City	2005 Sales	2006 Sales
NEW YORK	AUSTIN	2,599.00	2,780.37
	AUSTIN	860.00	958.79
	AUSTIN	1,774.00	2,071.25
	AUSTIN	1,484.00	1,773.50
	DALLAS	3,034.00	3,615.85
	DALLAS	1,034.00	1,109.97
NEW YORK		10,785.00	12,309.73
SAN FRAN	AUSTIN	6,094.00	6,262.50
	AUSTIN	1,346.00	1,597.62
	DALLAS	6,235.00	7,100.00
SAN FRAN		13,675.00	14,960.12
		24,460.00	27,269.85

### Trick 6:

Add the company Logo to the title.

```

ods escapechar = '^';
ods pdf file='c:\Lisbon.pdf';
title f=arial c=blue height=6 'Revenue Generated after 2004';
title2 ' ';
title3 j=c "^S={preimage='c:\ben\qp\SAS_logo.bmp'}";

```


#### Program 6a.

In this program, we are using an ODS statement to define an escape character. This is then used in the third title statement. Notice the syntax of the TITLE statements.

```
proc print data=sas_1.mdv ;  
    where city=: 'SAN FRAN';  
    id city;  
    var origcity;  
    var month /style={foreground=red};  
    sum sales2005 _4cast2006;  
  
run;  
ods pdf close;
```

#### Program 6b.

Also notice the STYLE= option on the VAR statement. If a column needs to be in a different color, one way to do it is to use the above syntax, ie a separate VAR statement for the variable you want to stand out.

Revenue Generated after 2004				
		Alliance Affiliate Member		
CITY	ORIGCITY	MONTH	Sales2005	_4cast2006
SAN FRANCISCO	LONDON	2	894.00	907.74
SAN FRANCISCO	LONDON	4	2,081.00	2,345.65
SAN FRANCISCO	SYDNEY	7	4,919.00	5,442.21
SAN FRANCISCO	SYDNEY	12	2,346.00	2,670.02
			10,240.00	11,365.62

#### Output 6.

### CONCLUSIONS

Well, I did not make it to 22 Tricks in the paper, but will provide more tricks during the presentation. Hopefully you, the reader, have seen that with the SAS System and its ODS component, one can create very colorful reports.

This paper has only scratched the surface with regard to what you can do with ODS. For instance, the topic of PROC TEMPLATE was purposely avoided so that this paper could fill the need of users who want a quick and easy way to add spice to their reports without going into a lot of detail. ODS is such a powerful and broad topic, that one could

spend days reading and learning more about this topic. For further information on ODS, go to [www.sas.com](http://www.sas.com) where you will find many books, courses and other resources provided by SAS Institute to help you on your ODS journey.

## **ACKNOWLEDGMENTS**

I would like to acknowledge and greatly thank the Technical Support Department at SAS Institute for their helpful knowledge and expertise that they so freely gave.

## **CONTACT INFORMATION**

Your comments and questions are valued and encouraged. Contact the author at:

Ben Cochran  
The Bedford Group  
Raleigh, NC 27607  
Office: 919.831.1191  
Fax: 919.831.1191  
Email: [bedfordgroup@nc.rr.com](mailto:bedfordgroup@nc.rr.com)  
Web: [www.bedford-group.com](http://www.bedford-group.com)

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.